NAG Library Routine Document

C09CAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C09CAF computes the one-dimensional discrete wavelet transform (DWT) at a single level. The initialization routine C09AAF must be called first to set up the DWT options.

2 Specification

```
SUBROUTINE C09CAF(N, X, LENC, CA, CD, ICOMM, IFAIL)
INTEGER           N, LENC, ICOMM(100), IFAIL
double precision  X(N), CA(LENC), CD(LENC)
```

3 Description

C09CAF computes the one-dimensional DWT of a given input data array, x_i, for $i = 1, 2, \ldots, n$, at a single level. For a chosen wavelet filter pair, the output coefficients are obtained by applying convolution and downsampling by two to the input, x. The approximation (or smooth) coefficients, C_a, are produced by the low pass filter and the detail coefficients, C_d, by the high pass filter. To reduce distortion effects at the ends of the data array, several end extension methods are commonly used. Those provided are: periodic or circular convolution end extension, half-point symmetric end extension, whole-point symmetric end extension or zero end extension. The number n_c, of coefficients C_a or C_d is returned by the initialization routine C09AAF.

4 References

Daubechies I (1992) *Ten Lectures on Wavelets* SIAM, Philadelphia

5 Parameters

1: N – INTEGER *Input*

On entry: the number of elements, n, in the data array x.

Constraint: This must be the same as the value N passed to the initialization routine C09AAF.

2: X(N) – ***double precision*** array *Input*

On entry: X contains the input dataset x_i, for $i = 1, 2, \ldots, n$.

3: LENC – INTEGER *Input*

On entry: the dimension of the arrays CA and CD as declared in the (sub)program from which C09CAF is called. This must be at least the number, n_c, of approximation coefficients, C_a, and detail coefficients, C_d, of the discrete wavelet transform as returned in NWC by the call to the initialization routine C09AAF.

Constraint: LENC $\geq n_c$, where n_c is the value returned in NWC by the call to the initialization routine C09AAF.

4: CA(LENC) – ***double precision*** array *Output*

On exit: CA(i) contains the ith approximation coefficient, $C_a(i)$, for $i = 1, 2, \ldots, n_c$.

5: CD(LENC) – ***double precision*** array *Output*

On exit: CD(i) contains the ith detail coefficient, $C_d(i)$, for $i = 1, 2, \ldots, n_c$.

6: ICOMM(100) – INTEGER array *Communication Array*

On entry: contains details of the discrete wavelet transform and the problem dimension as setup in the call to the initialization routine C09AAF.

On exit: contains additional information on the computed transform.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N is inconsistent with the value passed to the initialization routine C09AAF.

IFAIL $= 2$

On entry, LENC $< n_c$, where n_c is the value returned in NWC by the call to the initialization routine C09AAF.

IFAIL $= 6$

On entry, the initialization routine C09AAF has not been called first or it has been called with WTRANS $=$ 'M', or the communication array ICOMM has become corrupted.

7 Accuracy

The accuracy of the wavelet transform depends only on the floating point operations used in the convolution and downsampling and should thus be close to machine precision.

8 Further Comments

None.

9 Example

This example computes the one-dimensional discrete wavelet decomposition for 8 values using the Daubechies wavelet, WAVNAM $=$ 'DB4', with zero end extension.

9.1 Program Text

```
*     C09CAF Example Program Text
*     Mark 22 Release. NAG Copyright 2007.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NWCMAX
      PARAMETER        (NMAX=30,NWCMAX=30)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, N, NF, NWC, NWL, NY
      CHARACTER*12     MODE, WAVNAM, WTRANS
*     .. Local Arrays ..
      DOUBLE PRECISION CA(NWCMAX), CD(NWCMAX), X(NMAX), Y(NMAX)
      INTEGER          ICOMM(100)
*     .. External Subroutines ..
      EXTERNAL         C09AAF, C09CAF, C09CBF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C09CAF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
*     Read problem parameters.
      READ (NIN,*) N
      READ (NIN,*) WAVNAM, MODE
*
      IF (N.LE.NMAX .AND. N.GE.2) THEN
         WRITE (NOUT,*) 'DWT :: Wavelet: ', WAVNAM, ', End mode: ', MODE
*        Read array
         READ (NIN,*) (X(I),I=1,N)
         WRITE (NOUT,*) 'Input Data      X :'
         WRITE (NOUT,99997) (X(I),I=1,N)
*        Query wavelet filter dimensions
         WTRANS = 'Single Level'
         IFAIL = 1
*
         CALL C09AAF(WAVNAM,WTRANS,MODE,N,NWL,NF,NWC,ICOMM,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            IFAIL = 1
*
            CALL C09CAF(N,X,NWCMAX,CA,CD,ICOMM,IFAIL)
*
            IF (IFAIL.EQ.0) THEN
               WRITE (NOUT,99998)
               WRITE (NOUT,99997) (CA(I),I=1,NWC)
               WRITE (NOUT,99996)
               WRITE (NOUT,99997) (CD(I),I=1,NWC)
            ELSE
               WRITE (NOUT,*)
               WRITE (NOUT,99999) IFAIL
            END IF
*
            NY = N
*
            IFAIL = 1
*
            CALL C09CBF(NWC,CA,CD,NY,Y,ICOMM,IFAIL)
*
            IF (IFAIL.EQ.0) THEN
               WRITE (NOUT,99995)
               WRITE (NOUT,99997) (Y(I),I=1,NY)
            ELSE
               WRITE (NOUT,*) 'Error in C09CBF: IFAIL = ', IFAIL
            END IF
*
         ELSE
            WRITE (NOUT,*)
            WRITE (NOUT,99994) IFAIL
         END IF
*
      END IF
```

```
*
99999 FORMAT (1X,' ** C09CAF returned with IFAIL = ',I5)
99998 FORMAT (1X,' Approximation coefficients CA : ')
99997 FORMAT (1X,8(F8.4,1X),:)
99996 FORMAT (1X,' Detail coefficients        CD : ')
99995 FORMAT (1X,' Reconstruction              Y : ')
99994 FORMAT (1X,' ** C09AAF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
C09CAF Example Program Data
8              : N
DB4  Zero      : WAVNAM, MODE
1.0
3.0
5.0
7.0
6.0
4.0
5.0
2.0            : X(1:N)
```

9.3 Program Results

```
 C09CAF Example Program Results
 DWT :: Wavelet: DB4         , End mode: Zero
 Input Data      X :
   1.0000   3.0000   5.0000   7.0000   6.0000   4.0000   5.0000   2.0000
  Approximation coefficients CA :
   0.0011  -0.0043  -0.0174   4.4778   8.9557   7.3401   2.5816
  Detail coefficients        CD :
   0.0237   0.0410  -0.5966   1.7763  -0.7517   0.3332  -0.1188
  Reconstruction              Y :
   1.0000   3.0000   5.0000   7.0000   6.0000   4.0000   5.0000   2.0000
```